The Essentials of SAS® Dates and Times
Derek Morgan, Covidien

ABSTRACT
The first thing you need to know is that SAS® software stores dates and times as numbers. However, this is not the only thing that you need to know, and this presentation will give you a solid base for working with dates and times in SAS. It will also introduce you to functions and features that will enable you to manipulate your dates and times with surprising flexibility. This paper will also show you some of the possible pitfalls with dates (and times and datetimes) in your SAS code, and how to avoid them. We'll show you how the SAS System handles dates and times through examples, including the ISO 8601 formats and informats (new as of SAS 8.2), how to use dates and times in TITLE and/or FOOTNOTE statements, and close with a brief discussion of Excel conversions.

WHAT'S THE FIRST THING I NEED TO KNOW?
The first thing is, of course, that the SAS System stores dates, times and datetimes as numbers. Dates are counted in days from a zero point of January 1, 1960. Times are counted in seconds from a zero point of midnight of the current day, and datetimes are counted in seconds since midnight, January 1, 1960. Each day that passes increments the day counter by one, and each second that passes increments the time and datetime counters by one. This makes it easy to calculate durations in days and seconds. Unfortunately, most references to dates and times do not use the lowest common denominator of days and seconds, respectively, and they certainly don't use January 1, 1960, and midnight as their central references. That is where the first problem comes up: how to get SAS to speak about dates and times the way we do. How do you tell SAS that the date is January 14, 1967?

```sas
DATA _NULL_;
date1 = "January 14, 1967";
date2 = "October 23, 2006";
days_in_between = date2 - date1;
PUT days_in_between = ;
RUN;
```

That will not get you very far. Depending on the context, you will get an error message telling you that you tried to put characters into a numeric value, or you will get a character variable with the words, “January 14, 1967” stored in it. It may look okay, but if you try to do a calculation using that character variable, you will get the SAS equivalent of a blind date – a missing value.

```sas
NOTE: Character values have been converted to numeric values at the places given by:
(Line):(Column). 19   4:27
NOTE: Invalid numeric data, date2='October 23, 2006' , at line 4 column 19.
NOTE: Invalid numeric data, date1='January 14, 1967' , at line 4 column 27.

days_in_between= .
```

In order to tell SAS about a specific date, you use a “date literal.” The date literals for the two dates above are “14JAN1967” and “23OCT2006”. The letter “d” at the end tells SAS that this is a date, not a string of characters, so the code becomes:

```sas
DATA _NULL_;
date1 = '14jan1967'd;
date2 = '23oct2006'd;
days_in_between = date2 - date1;
PUT days_in_between = ;
RUN;
```

No part of the date literal is case-sensitive, that is, you can use all capital letters, all lower-case, or mixed case for the date inside the quotes and the “d” can be upper or lower-case. You may use single or double quotes to enclose the literal string, but if you use double quotes, you will be subject to macro variable resolution, which means that an ampersand (&) may cause unexpected results. Time and datetime literals are expressed in a similar fashion; however, instead of the letter “d”, they are followed by the letter “t” for time, or the letters “dt” for datetimes. Time
literals are expressed with a twenty-four hour clock in the form "05:00:00"t, and datetime literals are expressed as "23oct2006:05:00:00"dt.

SAVING SPACE BY SHRINKING VARIABLE LENGTHS

While SAS has a default length for its numeric variables of eight, you can save space by defining smaller lengths for dates, times, and datetimes. Dates can be stored in a length of four. Times can be stored in a length of four, unless you need decimal fractions of seconds; then you would use eight for maximum precision. Datetimes can safely be stored in a length of six, unless you need decimal fractions of seconds, in which case you would again use eight. For techies, if your operating system doesn’t handle half-words, use eight for datetimes. Why can’t you go any lower?

Given a date of August 4, 2006, if we run the following code, you will see:

```sas
DATA date_length;
  LENGTH len3 3 len4 4 len5 5;
  len3 = "04AUG2006"d + 2;
  len4 = len3;
  len5 = len3;
  FORMAT len3 len4 len5 mmddyy10.;
RUN;
```

Now let’s look at our data set:

While it isn’t the missing value dot, you can see that the value of len3 is not correct. When the numeric date value was written to the dataset, some precision was lost. This is a hit-or-miss proposition; sometimes it happens and sometimes it doesn’t. Do not take the risk.

OLD DATES

SAS can go all the way back to January 1, 1582, so you will likely be able to work with your old dates. However, old dates have the potential to produce incorrect dates in SAS. You may not get a missing value, but make sure that you check your century. The YEARCUTOFF option gives you the capability to define a 100-year range for two-digit year values. The default value for the YEARCUTOFF option is 1920, giving you a range of 1920-2019. Let’s demonstrate with date literals using this program:

```sas
OPTIONS YEARCUTOFF=1920; /* SAS System default */
DATA yearcutoff1;
  yearcutoff = "SAS System Default: 1920";
  date1 = "08AUG06"d;
  date2 = "15JUN48"d;
  date3 = "04jan69"d;
  date4 = "22oct95"d;
RUN;

OPTIONS YEARCUTOFF=1840;
DATA yearcutoff2;
  yearcutoff = "1840";
  date1 = "08AUG06"d;
  date2 = "15JUN48"d;
  date3 = "04jan69"d;
  date4 = "22oct95"d;
RUN;
```
The above code gives us the following result:

<table>
<thead>
<tr>
<th>OPTIONS</th>
<th>date1</th>
<th>date2</th>
<th>date3</th>
<th>date4</th>
</tr>
</thead>
<tbody>
<tr>
<td>YEARCUTOFF value</td>
<td>08/08/2006</td>
<td>06/15/1948</td>
<td>01/04/1969</td>
<td>10/22/1995</td>
</tr>
<tr>
<td>SAS System Default: 1920</td>
<td>08/08/1906</td>
<td>06/15/1848</td>
<td>01/04/1869</td>
<td>10/22/1895</td>
</tr>
<tr>
<td>1840</td>
<td>08/08/1906</td>
<td>06/15/1848</td>
<td>01/04/1869</td>
<td>10/22/1895</td>
</tr>
</tbody>
</table>

Any two-digit year that the SAS System has to translate, whether it is from a date literal as shown in the above example, an ASCII file being processed with the INPUT statement and an informat or even the INPUT() function and an informat will be affected by this option. The lesson here is to check your dates before and after processing.

FORMATS AND TRANSLATING SAS DATES

Since SAS keeps track of dates and times (and datetimes) as numbers relative to some fixed point in time, how do we get SAS to show us its dates in ways that we understand, and how can we communicate our dates to SAS? Formats are the way that SAS can translate what it understands into something that we can understand, while informats do the reverse. So how can this built-in translation fail?

First, you need to make sure you are using the correct format or informat for your data, and the type of data you are working with. Do not try to use a date format to print out a datetime value, or use a time format to print out a date. SAS stores dates, times, and datetimes as numbers, but it does not store any context information with it.

Unfortunately, this means that if it is not clear what the value represents to you, SAS will not be much help directly. (You can make an educated guess based on the maximum values and ranges of the variables involved, but that method is not foolproof, and it would be data-dependent.) Here’s a little program to illustrate:

```sas
DATA _NULL_;
  date = "08AUG2003"d;
  time = "13:43"t;
  datetime = "25JAN2005:15:52:07"dt;
  PUT "MMDDYY10. representation of date=" date mmddyy10. /
       "MONYY7. representation of date=" date monyy7. /
       "TIMEAMPM9. representation of date=" date timeampm9. /
       "DTMONYY7. representation of date=" date dtmonyy. /
       "When value of date is used as a SAS *datetime* value, the date represented is:" date
datetime20. /
       "DATETIME20. representation of time=" time datetime20. /
       "DTMONYY7. representation of time=" time dtmonyy7. /
       "TIMEAMPM9. representation of time=" time timeampm9. /
       "MONYY7. representation of time=" time monyy7. /
       "When value of time is used as a SAS *date* value, the date represented is:" time
      mmddyy10. /
       "DATETIME20. representation of datetime=" datetime datetime20. /
       "DTMONYY7. representation of datetime=" datetime dtmonyy7. /
       "TIMEAMPM9. representation of datetime=" datetime timeampm9. /
       "MONYY7. representation of datetime=" datetime monyy7. /
       "When value of datetime is used as a SAS *date* value, the date represented is:"
datetime mmddyy10.;
RUN;
```

To make it a little easier to compare and contrast, here are the results in tabular form. Any DATA step and REPORT or TABULATE procedure manipulations necessary to produce this output are left as an exercise for the reader.

<table>
<thead>
<tr>
<th>Variable</th>
<th>Value in SAS</th>
<th>Date Formats</th>
<th>Time Format</th>
<th>Datetime Formats</th>
</tr>
</thead>
<tbody>
<tr>
<td>Date</td>
<td>15925</td>
<td>08/08/2003</td>
<td>4:25:25 AM</td>
<td>01JAN1960</td>
</tr>
<tr>
<td></td>
<td></td>
<td>AUG2003</td>
<td>JAN1960</td>
<td>01JAN1960:04:25:25</td>
</tr>
<tr>
<td>Time</td>
<td>49380</td>
<td>03/13/2095</td>
<td>1:43:00 PM</td>
<td>JAN1960</td>
</tr>
<tr>
<td></td>
<td></td>
<td>MAR2095</td>
<td>JAN1960</td>
<td>01JAN1960:13:43:00</td>
</tr>
<tr>
<td>Datetime</td>
<td>1422287527</td>
<td>********</td>
<td>3:52:07 PM</td>
<td>JAN2005</td>
</tr>
</tbody>
</table>
The first thing you should notice is that the datetime value gives you several asterisks when you try to format it as a date. The date represented by the value 1,422,287,527 is so far in the future that it cannot be represented by a four-digit year, but that is the only blatant indication that something’s not quite right. Why is there a discrepancy on the others? When you try to translate a date value with a time format, you are translating days since January 1, 1960 using something designed to translate seconds since midnight. 15,925 seconds after midnight is 4:25:25 in the morning. If you translate 15,925 as seconds after midnight of January 1, 1960, which is the datetime convention, you get 4:25:25 AM on January 1, 1960. Similarly, if you translate 49,380 as days since January 1, 1960, you get March 13, 2095. Finally, note the cell in italics. There is absolutely nothing to indicate that something is wrong here. Why do we get a normal-looking time? The TIMEAMPM. format gives times from 12:00 AM to 11:59 PM, so any value greater than 86,400 (the number of seconds in a day) just cycles into the next day. Therefore, you are getting the result of the calculation MOD(1422287527,86400), which is 57,127, and translates into a time of 3:52:07 PM using the time scale of seconds after midnight.

NEED A FORMAT FOR YOUR DATE?

Although there are many formats built into the SAS System, you may find yourself in a position where you cannot find a format that displays your date, time, or datetime the way you want. Don’t panic. You can create and use a custom format to show off your dates. There are two ways to do this and they both require using the FORMAT procedure. The first way uses the VALUE statement. You define a range for the values using date, time, or datetime constants, and then you can tell SAS what to print out instead of the date. Here’s a sample program that will create a format to display whether a contract is scheduled for arbitration or renegotiation based on the expiration date of the contract:

```sas
PROC FORMAT LIBRARY=library;
VALUE contrct
 LOW-'31dec2005'd="INVALID"
 '01JAN2006'd-'31JUL2006'd= "ARBITRATION"
 '01AUG2006'd - '31DEC2006'd = "RENEGOTIATION"
 '01JAN2007'd - HIGH=[MONYY7.];  /* INSTRUCTS SAS TO USE THE
                  MONYY7. FORMAT FOR VALUES BEYOND 2006*/
RUN;

PROC PRINT DATA= contracts;
ID contract_num;
VAR exp_date exp_date_raw;
FORMAT exp_date contrct. exp_date_raw MMDDYY10.;
RUN;
```

Here is some of the output–instead of printing the date for the variable EXP_DATE, our format classifies the date values and translates them into categorical text.

<table>
<thead>
<tr>
<th>contract_num</th>
<th>exp_date</th>
<th>exp_date_raw</th>
</tr>
</thead>
<tbody>
<tr>
<td>5829014</td>
<td>INVALID</td>
<td>12/06/2005</td>
</tr>
<tr>
<td>9330471</td>
<td>RENEGOTIATION</td>
<td>09/21/2006</td>
</tr>
<tr>
<td>6051271</td>
<td>APR2007</td>
<td>04/11/2007</td>
</tr>
<tr>
<td>2301911</td>
<td>ARBITRATION</td>
<td>01/23/2006</td>
</tr>
<tr>
<td>6894300</td>
<td>RENEGOTIATION</td>
<td>08/21/2006</td>
</tr>
</tbody>
</table>

So where can you go wrong here? Several places, actually. Let’s examine the code for our format:

```sas
1 PROC FORMAT LIBRARY=library;
2 VALUE contract
3 LOW='31dec2005'd='INVALID'
4  '01JAN2006'd-'31JUL2006'd= 'ARBITRATION'
5  '01AUG2006'd - '31DEC2006'd = 'RENEGOTIATION'
6  '01JAN2007'd - HIGH=[monyy7.];  /* INSTRUCTS SAS TO USE THE MONYY7. FORMAT FOR VALUES BEYOND 2006*/
7 RUN;
```

First, if you forget the "d" to indicate that the value is a date constant, you are going to get an error from lines 3-6. Notice that line 3 uses the special value “LOW”. Without it, any date before January 1, 2006, will display as the actual
SAS numeric value. Similarly, line 6 accounts for values in the future by using the special value “HIGH”. However, instead of setting it to print categorical text, we have told SAS to use one of its own date formats if the date is after December 31, 2006. That is why there is a format name enclosed in brackets after the equal sign. Without the format name, there would be no formatting associated with the SAS date value, and all you would see printed out would be the number of days since January 1, 1960.

PRETTY AS A PICTURE
The second way to create your own format for your date, time, or datetime is with a picture format. Picture formats allow you to create a representation of your data by describing what you want it to look like. There are special formatting directives to allow you to represent dates, times and datetime values. These directives are case-sensitive. You will also need to use the DATATYPE= option in your PICTURE statement. DATATYPE is DATE, TIME, or DATETIME to indicate the type of value you are formatting.

Here are the directives:

<table>
<thead>
<tr>
<th>%a</th>
<th>Locale’s abbreviated weekday name.</th>
</tr>
</thead>
<tbody>
<tr>
<td>%A</td>
<td>Locale’s full weekday name.</td>
</tr>
<tr>
<td>%b</td>
<td>Locale’s abbreviated month name.</td>
</tr>
<tr>
<td>%B</td>
<td>Locale’s full month name.</td>
</tr>
<tr>
<td>%d</td>
<td>Day of the month as a decimal number (1-31), with no leading zero. Put a zero between the percent sign and the &quot;d&quot; to have a leading zero in the display.</td>
</tr>
<tr>
<td>%H</td>
<td>Hour (24-hour clock) as a decimal number (0-23), with no leading zero. Put a zero between the percent sign and the &quot;H&quot; to have a leading zero in the display.</td>
</tr>
<tr>
<td>%I</td>
<td>Hour (12-hour clock) as a decimal number (1-12), with no leading zero. Put a zero between the percent sign and the &quot;I&quot; to have a leading zero in the display.</td>
</tr>
<tr>
<td>%j</td>
<td>Day of the year as a decimal number (1-366), with no leading zero. Put a zero between the percent sign and the &quot;j&quot; to have a leading zero in the display.</td>
</tr>
<tr>
<td>%m</td>
<td>Month as a decimal number (1-12), with no leading zero. Put a zero between the percent sign and the &quot;m&quot; to have a leading zero in the display.</td>
</tr>
<tr>
<td>%M</td>
<td>Minute as a decimal number (0-59), with no leading zero. Put a zero between the percent sign and the &quot;M&quot; to have a leading zero in the display.</td>
</tr>
<tr>
<td>%p</td>
<td>Either AM or PM.</td>
</tr>
<tr>
<td>%S</td>
<td>Second as a decimal number (0-59), with no leading zero. Put a zero between the percent sign and the &quot;S&quot; to have a leading zero in the display.</td>
</tr>
<tr>
<td>%U</td>
<td>Week number of the year (Sunday as the first day of the week) as a decimal number (0-53), with no leading zero. Put a zero between the percent sign and the &quot;U&quot; to have a leading zero in the display.</td>
</tr>
<tr>
<td>%W</td>
<td>Weekday as a decimal number, where 1 is Sunday, and Saturday is 7.</td>
</tr>
<tr>
<td>%Y</td>
<td>Year without century as a decimal number (0-99), with no leading zero. Put a zero between the percent sign and the &quot;y&quot; to have a leading zero in the display.</td>
</tr>
<tr>
<td>%%</td>
<td>The percent character (%).</td>
</tr>
</tbody>
</table>

Here is a simple example of using the date directives to create an enhanced date display with the day of the year

```sas
PROC FORMAT;
   PICTURE xdate
   . -.z = "No Date Given"
   LOW - HIGH = ‘%B %d, %Y is day %j of %Y’ (DATATYPE=DATE);
RUN;
PROC PRINT DATA=pictest;
   VAR date;
   FORMAT date xdate40.;
RUN;
```
Let's look at the output for several pseudo-random dates:

<table>
<thead>
<tr>
<th>date</th>
</tr>
</thead>
<tbody>
<tr>
<td>No Date Given</td>
</tr>
<tr>
<td>August 3, 2005 is day 215 of 2005</td>
</tr>
<tr>
<td>November 27, 2005 is day 331 of 2005</td>
</tr>
<tr>
<td>January 14, 2008 is day 14 of 2008</td>
</tr>
<tr>
<td>October 6, 2007 is day 279 of 2007</td>
</tr>
<tr>
<td>February 13, 2006 is day 44 of 2006</td>
</tr>
<tr>
<td>May 18, 2007 is day 138 of 2007</td>
</tr>
<tr>
<td>August 12, 2007 is day 224 of 2007</td>
</tr>
<tr>
<td>October 2, 2005 is day 275 of 2005</td>
</tr>
<tr>
<td>March 22, 2007 is day 81 of 2007</td>
</tr>
</tbody>
</table>

Well, this is impressive. Where can you go wrong with this? First, make sure that you have defined the correct DATATYPE. Otherwise, you are not going to get a correct representation, just like with the built-in SAS formats. Second, you need to make sure that you use a length that is long enough to show all of your text. The default length of a picture format is the number of characters between the quotes in the picture. However, there may be more characters in your output. That is why the format length in the FORMAT statement of the PRINT procedure is set to 40. If you just leave the format at its default length, you would see this output:

<table>
<thead>
<tr>
<th>date</th>
</tr>
</thead>
<tbody>
<tr>
<td>August 3, 2005 is day 215</td>
</tr>
<tr>
<td>November 27, 2005 is day 331</td>
</tr>
<tr>
<td>January 14, 2008 is day 1</td>
</tr>
<tr>
<td>October 6, 2007 is day 279</td>
</tr>
<tr>
<td>February 13, 2006 is day 44</td>
</tr>
<tr>
<td>May 18, 2007 is day 138</td>
</tr>
<tr>
<td>August 12, 2007 is day 224</td>
</tr>
<tr>
<td>October 2, 2005 is day 275</td>
</tr>
<tr>
<td>March 22, 2007 is day 81</td>
</tr>
</tbody>
</table>

Sadly, none of these match the desired formatting in the previous output, because the default length of the format is set by the maximum length of the text in the PICTURE statement in the code above, which is 25 characters. That is not long enough to accommodate all of the text in the format because each of the format directives are only two characters long, although the values they display can be longer than that.

**DATES IN TITLES AND FOOTNOTES**

Now that we know how to dress up our dates just the way we want them, how can we show them other than in the detail of our reports? For example, if you have a report that is run every week, you could put the date in the title like this:

```
TITLE 'Date of report: March 24, 2009';
```

Unfortunately, that means you will be responsible for changing the code every month. You can get around this by using one of the four automatic macro variables in the SAS system: &SYSDATE; &SYSDATE9; &SYSDAY, &SYSTIME. They are set when the SAS job starts, and you cannot change them. If you want to use one of these variables, this is how you would do it:

```
TITLE "Date of report: &SYSDATE9"; /* Since you are using a macro variable, you MUST have DOUBLE quotes around the text */
```
If you were to run this job on March 24, 2009, this statement would put the text “Date of report: 24MAR2009” at the top of each page. On March 25, it would be “Date of report: 25MAR2009”. That is functional, but not very appealing. None of the macro variables is particularly appealing in their native format: &SYSTIME comes out as a 24-hour clock value (e.g., 23:00), while &SYSDATE is the same as &SYSDATE9 with a two-digit year (24MAR09). However, &SYSDAY will look like a proper day of the week (Tuesday).

If that’s not exactly what you had in mind, don’t worry. You can take advantage of formats and display dates (and times) within your TITLES and FOOTNOTES exactly how you want them to look. You can always get the current date and time from SAS using the DATE(), TIME(), and/or DATETIME() functions. It will involve the creation of a macro variable to hold your text, but it takes only a little macro or DATA step coding to do it. Both methods use either the PUT() or PUTN() functions to turn the SAS date value into a text string.

**Using a DATA Step and CALL SYMPUT() to Create your Macro Variable**

```sas
DATA _NULL_; /* Don't need to create a dataset, just execute DATA step code
CALL SYMPUT('rdate',LEFT(PUT(DATE(),worddate32.))); /* Line 2 creates a global macro variable called &RDATE and gives it the value of
today's date formatted with the worddate32. format. Use the LEFT() function to
remove leading spaces or else you'll get an unwelcome surprise! */
RUN;
TITLE "Date of report: &rdate"; /* Don't forget DOUBLE quotes! */
```

The value of the macro variable &RDATE is “March 24, 2009”, and it is left-justified, so the title on each page will now read “Date of report: March 24, 2009”. You can take this code as written above, change the format from WORDDATE32. to whatever you need, put it into your reports and your dates will automatically change each day they are run.

**The Fancy Example Using Custom Formats and Macro Functions**

This will show what you can do with custom formats and how you can put them into TITLES and FOOTNOTES using SAS macro functions. Once the format is created, this can also be done with a DATA step as shown above. The first part of the example below creates a custom format named DEDATE using the PICTURE statement.

```sas
PROC FORMAT;
PICTURE dedate
. . z = "No Date Available" /* What if the date (datetime in this case) is
missing? */
LOW - HIGH = '%B %d, %Y at %I:%0M %p' (DATATYPE=DATETIME)
;
/* Now we use the %SYSFUNC() and %QSYSFUNC() functions to get access to DATA step
functions in the macro language */
%LET rdate=%LEFT(%QSYSFUNC(PUTN(%SYSFUNC(DATETIME()),dedate32.)));
TITLE "Date of report: &rdate"; /* Don't forget DOUBLE quotes! */
```

The FORMAT procedure uses a mixture of text and date directives to create the display. Line 3 is there in case a datetime value is missing (if you use the DATETIME() function, it will never be missing.) Line 4 contains the date directives as well as text that will be printed with the date directives, but the most important part of the line is the DATATYPE= argument. This is not optional, because it tells the format what type of value to expect so that it can be translated correctly. The value of the DATATYPE argument can be DATE, TIME, or DATETIME. Sending the wrong type of data to a custom format will give you incorrect results just like sending the wrong type of data to a built-in SAS format does.

To decipher line 8, we need to work from the inside of the parentheses out, so the first thing we do is get the current SAS datetime value with %SYSFUNC(DATETIME()). We use the %SYSFUNC() function because it allows us to use a SAS function somewhere other than inside of a DATA step. Next, we want to put that numeric value into a text string, which adds the %QSYSFUNC (because the format name contains a character that may mean something to the macro language processor) to call the PUTN() function. This gives us: %QSYSFUNC(PUTN(%SYSFUNC(DATETIME()),dedate32.)). You need to specify the length of the format because its default length is only 22 (the number of characters between the quotes in line 4.) Finally, remember that SAS right-justifies numbers by default, so we use the SAS autocall macro %LEFT() to left-justify the result and store it in the macro variable &RDATE. Our report title will now say, Date of report: March 24, 2009 at 4:08 PM (italics mine, not actual appearance) as per the date directive in line 4. The only caution is that your title will be updated each time you execute the code that creates the macro variable. If you do not want the title line to update throughout your report, make sure you only execute the code once at the beginning of your job.
READING DATES AND TIMES AND DATETIMES

So far, our examples have all used date constants, but you cannot put a date constant everywhere you need a date, such as in data. If you are converting data from a flat file, then you will need to use informats to read the data. You will need both the formatted INPUT statement and an informat in order to read date, time, or datetime data from a flat file. Here is an example of a flat file with dates:

```
10/26/2000
09/03/1998
05/14/1967
08/25/1989
07/01/2004
03/16/2001
03/16/1971
04/03/1968
09/25/1965
```

To read the above file, you would use this DATA step:

```
DATA read_dates;
INFILE "c:\book\examples\a_few_dates.txt";
INPUT @1 sample_date MMDDYY10.;
RUN;
```

Here is the log of a program to display what is stored.

```
DATA _null_;
SET read_dates;
PUT 'Sample date without formatting ' sample_date /
'Sample date with WORDDATE format ' sample_date WORDDATE.;
RUN;
```

Sample date without formatting 14909
Sample date with WORDDATE. format October 26, 2000
Sample date without formatting 14125
Sample date with WORDDATE. format September 3, 1998
Sample date without formatting 2690
Sample date with WORDDATE. format May 14, 1967
Sample date without formatting 10829
Sample date with WORDDATE. format August 25, 1989
Sample date without formatting 16253
Sample date with WORDDATE. format July 1, 2004
Sample date without formatting 15050
Sample date with WORDDATE. format March 16, 2001
Sample date without formatting 4092
Sample date with WORDDATE. format March 16, 1971
Sample date without formatting 3015
Sample date with WORDDATE. format April 3, 1968
Sample date without formatting 2094
Sample date with WORDDATE. format September 25, 1965
Since we looked at the file first, we knew that all of the data looked like “mm/dd/yyyy”, and we simply told the INPUT statement what it would see when it read the field. By specifying that informat, we told SAS how to translate what seems to be a character string (“/” is not a number) into a SAS date value. It’s easy to get the wrong result here: if you use the wrong informat for your data, things will definitely go wrong. In most cases, using the wrong informat will give you an error, but you need to be careful with some of the informats that differ only in the order of month, day, and/or year. The MMDDYY. informat will not give you the same result as the DDMMYY. informat, and it would not give you any message that anything was abnormal until the middle two digits in the data field were greater than 12.

Let’s see what happens when we use the wrong informat with the same file:

```plaintext
1   DATA read_dates;
2   INFILE 'c:\book\examples\a_few_dates.txt';
3   INPUT @1 sample_date DDMMYY10.;
4   RUN;

NOTE: Invalid data for sample_date in line 1 1-10.
RULE:     ----+----1----+----2----+----3----+----4----+----5----+----6----+----7----+----8----+--
1         10/26/2000 10
sample_date=. _ERROR_=1 _N_=1
NOTE: Invalid data for sample_date in line 3 1-10.
3         05/14/1967 10
sample_date=. _ERROR_=1 _N_=3
NOTE: Invalid data for sample_date in line 4 1-10.
4         08/25/1989 10
sample_date=. _ERROR_=1 _N_=4
NOTE: Invalid data for sample_date in line 6 1-10.
6         03/16/2001 10
sample_date=. _ERROR_=1 _N_=6
NOTE: Invalid data for sample_date in line 7 1-10.
7         03/16/1971 10
sample_date=. _ERROR_=1 _N_=7
NOTE: Invalid data for sample_date in line 9 1-10.
9         09/25/1965 10
sample_date=. _ERROR_=1 _N_=9
```

At least you can see the error messages. But you did not get an error message on every line. What happened to the data on lines 2, 5 and 8?

<table>
<thead>
<tr>
<th>Sample date without formatting</th>
<th>Sample date with WORDDATE format</th>
</tr>
</thead>
<tbody>
<tr>
<td>.</td>
<td>.</td>
</tr>
<tr>
<td>Sample date with WORDDATE format</td>
<td>March 9, 1998 /* Not September 3, 1998 */</td>
</tr>
<tr>
<td>Sample date without formatting</td>
<td>.</td>
</tr>
<tr>
<td>Sample date with WORDDATE format</td>
<td>.</td>
</tr>
<tr>
<td>Sample date without formatting</td>
<td>.</td>
</tr>
<tr>
<td>Sample date with WORDDATE format</td>
<td>.</td>
</tr>
<tr>
<td>Sample date without formatting</td>
<td>.</td>
</tr>
<tr>
<td>Sample date with WORDDATE format</td>
<td>.</td>
</tr>
<tr>
<td>Sample date without formatting 13947</td>
<td></td>
</tr>
<tr>
<td>Sample date with WORDDATE format</td>
<td>January 7, 2004 /* Not July 1, 2004 */</td>
</tr>
<tr>
<td>Sample date without formatting</td>
<td>.</td>
</tr>
<tr>
<td>Sample date with WORDDATE format</td>
<td>.</td>
</tr>
<tr>
<td>Sample date without formatting</td>
<td>.</td>
</tr>
<tr>
<td>Sample date with WORDDATE format</td>
<td>.</td>
</tr>
<tr>
<td>Sample date without formatting  2985</td>
<td></td>
</tr>
<tr>
<td>Sample date with WORDDATE format</td>
<td>March 4, 1968 /* Not April 3, 1968 */</td>
</tr>
<tr>
<td>Sample date without formatting</td>
<td>.</td>
</tr>
<tr>
<td>Sample date with WORDDATE format</td>
<td>.</td>
</tr>
</tbody>
</table>

You can see how important using the correct informat is; instead of the data you want, you get missing values and incorrect data.
READING DATES AND TIMES AND DATETIMES THE EASY WAY

Version 9 of SAS addressed one major problem with reading dates from flat files into SAS: you had to know what your date and time data looked like before you could process it from a flat file into SAS. The ANYDTDTE., ANYDTDTM., and ANYDTTME. informats will translate dates, times, and datetime values, respectively, into their corresponding SAS System values without having to know the representation of these dates, datetime, and time values in advance.

Of course, there are a couple of warnings: if you are going to leave it to SAS to figure out what the data look like, it will add some overhead to the processing of your flat file. How much extra overhead is involved? Let’s read 10,000 date records with SAS for Windows XP on a PC and find out. Here is a partial log:

```
DATA test;
  INFILE 'examples/datefile.txt' PAD MISSOVER;
  INPUT @1 date mmddyy10.;
RUN;

NOTE: 10000 records were read from the infile "examples/datefile.txt".
The minimum record length was 10.
The maximum record length was 10.
NOTE: The data set WORK.TEST has 10000 observations and 1 variables.
NOTE: DATA statement used (Total process time):
    real time 0.03 seconds
    cpu time 0.01 seconds
```

Now here is the same file processed with the ANYDTDTE10. informat:

```
DATA test2;
  INFILE 'examples/datefile.txt' PAD MISSOVER;
  INPUT @1 date anydtdte10.;
RUN;

NOTE: 10000 records were read from the infile "examples/datefile.txt".
The minimum record length was 10.
The maximum record length was 10.
NOTE: The data set WORK.TEST2 has 10000 observations and 1 variables.
NOTE: DATA statement used (Total process time):
    real time 0.07 seconds
    cpu time 0.04 seconds
```

The ANYDTDTE. informat takes four times as long in CPU time on this relatively small flat file as the MMDDYY. format does. So why does it matter if it takes more time? Doesn’t it solve all of your problems? Unfortunately, no, it does not solve all your date processing problems. What if you have to read a value such as "06/11/2005"? Or even worse, "06/11/05"? Is it June 11, 2005, November 6, 2005, or November 5, 2006? There is an option that will allow you to tell SAS what it should be. The DATESTYLE option comes in seven flavors:

- **MDY**
  - Sets the default order as month, day, year.
  - "06-11-05" would be translated as June 11, 2006

- **YMD**
  - Sets the default order as year, month, day.
  - "06-11-05" would be translated as June 11, 2006

- **MYD**
  - Sets the default order as month, year, day.
  - "06-11-05" would be translated as June 11, 2006

- **DYM**
  - Sets the default order as day, month, year.
  - "06-11-05" would be translated as June 11, 2006

- **LOCALE**
  - Sets the default value according to the LOCALE= system option. When the default value for the LOCALE= system option is “English_US”, this sets DATESTYLE to MDY. Therefore, by default, "06-11-05" would be translated as June 11, 2006.

Although it may now seem simple to resolve ambiguous dates, it is not foolproof. Always check your data for missing values.
AM I READING ENOUGH?
One last issue with reading dates and times from a flat file is making sure that you are reading the correct number of
characters in the field. All of the informats have a length specification, which is how many characters are to be
processed using the informat. For example, the MMDDYY10. informat reads ten characters and will attempt to
translate it as mm-dd-yyyy, although the exact separator between month, day, and year may vary. If you forget the
length specification, each informat has its own default length. However, this default may not match your data.
The bottom line is that when you are reading in date values from a flat file, it is best to look at the data that you have
produced from the flat file before you actually use it.

CREATING DATES, TIMES AND DATETIMES FROM NUMBERS
There are three functions in the SAS System that create SAS date, time, and datetime values from individual
elements such as month, year, hour, and minute. The MDY() function takes individual values for month, day and
year, and converts them to a SAS date value. Similarly, the HMS() function creates SAS time values from hours,
minutes, and seconds. Finally, the DHMS() function yields SAS datetime values from dates, hours, minutes, and
seconds. Each of the functions needs all of its component arguments, e.g., the DHMS() function needs non-missing
values for date, hour, minute, and second. If you are missing any one of these components, you will get a missing
value as the result. All the arguments to these functions must be numeric; they can be actual values or numeric
variables. In the DHMS() function, the date used can be a date literal, a SAS date value, or a numeric variable.
Month, day and year are the arguments to the MDY() function as follows: MDY(month,day,year);
This function calculates a SAS date value as long as: 1) none of the arguments are missing; 2) month is between 1 and 12, and 3)
the combination of month, day, and year is a valid calendar date. As an example of the last, MDY(2,31,2008); will
return a missing value because February 31, 2008, does not exist.
The HMS(hours,minutes,seconds); function creates a SAS time value from the hours, minutes, and seconds given,
and once again, all three values must be non-missing. When you use this function, it is important to remember that
there are two distinct concepts of time: 1) clock time, and 2) elapsed time. Because of that, the only restriction on this
function is that hours must be greater than or equal to zero, while minutes and seconds have no restrictions on
their value. HMS(7,45,80); will calculate to 27,980 seconds; to the SAS System, that is exactly the same as
HMS(7,46,20). The format that you use ultimately determines the context of the time value. If you are talking about
clock time, the value returned will be MOD(result,86400). You can use the following program to test this and help
your understanding, substituting actual non-clock values for hours, minutes, and second. The first value printed in
the resulting log will be the total number of seconds; the second value will display the number of seconds calculated
to a 24-hour clock (that MOD(value,86400) thing again.)
\begin{verbatim}
DATA _NULL_;
x = HMS(hours,minutes,seconds);
PUT x= +8 'x=' x time.;
RUN;
\end{verbatim}
The DHMS(date,hour,minute,second); function will create a datetime value from the component parts of date,
hour, minute, and second in the same way the previous two functions work. Again, each argument must not be
missing. date can be a date literal, a SAS date value, or a numeric variable.

EXTRACTING MONTH, DAY, YEAR, ETC., FROM YOUR SAS DATE
There are also functions that will pull the individual component values from SAS date, time and datetime values. The
list is below, and arg represents the appropriate date, time or datetime value for the function.

\begin{tabular}{ll}
DATEPART(arg) & Extracts the date from a SAS System datetime value as a SAS System date value. \\
DAY(arg) & Extracts the number of the day of the month from a SAS System date value. \\
HOUR(arg) & Extracts the hour from a SAS System time value. \\
MINUTE(arg) & Extracts the minute from a SAS System time value. \\
MONTH(arg) & Extracts the month from a SAS System date value. \\
SECOND(arg) & Extracts the second from a SAS System time value. \\
TIMEPART(arg) & Extracts the time portion from a SAS System datetime value as a SAS System time value. \\
WEEKDAY(arg) & Extracts the number of the day of the week, where Sunday=1, Monday=2, etc. from a SAS System date value. \\
YEAR(arg) & Extracts the year from a SAS System date value. Since obtaining the year value is the point of
using this function, it is important to remember that the YEARCUTOFF option affects two-digit
years. \\
\end{tabular}

You need to make sure that you use the appropriate function for the type of value you are working with. For example,
do not try to use the DAY() function to extract a day from your datetime value. It will not work correctly, as the next
element shows.
DATA _NULL_;  
datetime = '27JAN1960:12:35:00'dt;  
time = '12:35:00't;  
date = '27JAN1960'd;  
day_datim = DAY(datetime);  
day_tm = DAY(time);  
day_date = DAY(date);  
PUT datetime= +3 time= +3 date= ;  
PUT day_datim= +3 day_tm= +3 day_date=;  
RUN;

datetime=2291700    time=45300    date=26  
   day_datim=20    day_tm=10    day_date=27

The first line of output shows the actual values that SAS is operating on. In the above example, I purposely used an early datetime value (2,291,700 seconds relative to midnight, January 1, 1960) because it gives me a non-missing value, albeit incorrect, for the day. If you use a datetime sometime after March of 1960, SAS will give you an error message. However, the error message does not occur because SAS recognizes it as a datetime value; rather it occurs because it is trying to determine a day for a value that is too large for the function to work. In that vein, here is a trickier one:

DATA _NULL_;  
datetime = '27MAR1998:12:15:00'dt;  
time = '12:15:00't;  
date = '27MAR1998'd;  
minit_datim = MINUTE(datetime);  
minit_tm = MINUTE(time);  
minit_date = MINUTE(date);  
PUT datetime= +3 time= +3 date= ;  
PUT minit_datim= +3 minit_tm= +3 minit_date=;  
RUN;

datetime=1206620100    time=44100    date=13965  
   minit_datim=15    minit_tm=15    minit_date=52

What happened here? Isn’t that the correct value for the datetime value? Not exactly. Remember that both times and datetimes are stored in seconds since midnight, and that times cycle every 86,400 seconds. Any value greater than 86,400 (the number of seconds in a day) just cycles into the next day. Therefore, when you calculate the minutes from the datetime value as shown above, although in reality, you are getting the result of MOD(value,86400). So while the time extraction functions seem to work with datetimes, it is not good practice. You can also clearly see that trying to get the minute out of the date value gives you an incorrect result.

The YEAR() function is affected by the YEARCUTOFF option only if you have a date literal with a two-digit year, and this can lead to unexpected results. On the other hand, if you are extracting the year from a SAS date value, the YEARCUTOFF option will have no effect, because the concept of two-digit year vs. four-digit year does not exist; it is simply the number of days since January 1, 1960.

INTERVALS

Intervals are another way people track periods of time. The SAS System can allow you to manipulate dates and times and datetimes by these intervals. SAS Intervals are very powerful, but frequently misunderstood, especially when they are used in association with the two interval calculation functions INTCK() and INTNX(). The INTCK() function counts the number of intervals between two given dates, times, or datetimes, while the INTNX() function takes a given date, time or datetime and increments it by a given number of intervals. The SAS System has several standard intervals defined in it. The intervals represent periods of time such as days, weeks, months and quarters, to name a few. A complete list of the SAS intervals, along with the default starting points for each type of interval, is on the following page.
<table>
<thead>
<tr>
<th>Interval Name</th>
<th>Definition</th>
<th>Default Starting Point</th>
</tr>
</thead>
<tbody>
<tr>
<td>DAY</td>
<td>Daily intervals</td>
<td>Each day</td>
</tr>
<tr>
<td>WEEK</td>
<td>Weekly intervals of seven days</td>
<td>Each Sunday</td>
</tr>
<tr>
<td>WEEKDAYdaysW</td>
<td>Daily intervals with Friday-Saturday-Sunday counted as the same day (five-day work week with a Saturday-Sunday weekend). days identifies the individual numbers of the weekend day(s) by number (1=Sunday ... 7=Saturday). By default, days=&quot;17&quot;, so the default interval is WEEKDAY17W.</td>
<td>Each day</td>
</tr>
<tr>
<td>TENDAY</td>
<td>Ten-day intervals (a U.S. automobile industry convention)</td>
<td>1st, 11th, and 21st of each month</td>
</tr>
<tr>
<td>SEMIMONTH</td>
<td>Half-month intervals</td>
<td>1st and 16th of each month</td>
</tr>
<tr>
<td>MONTH</td>
<td>Monthly intervals</td>
<td>1st of each month</td>
</tr>
<tr>
<td>QTR</td>
<td>Quarterly (three-month) intervals</td>
<td>1-Jan, 1-Apr, 1-Jul, 1-Oct</td>
</tr>
<tr>
<td>SEMIYEAR</td>
<td>Semi-annual (six-month) intervals</td>
<td>1-Jan, 1-Jul</td>
</tr>
<tr>
<td>YEAR</td>
<td>Yearly intervals</td>
<td>1-Jan</td>
</tr>
<tr>
<td>DTDAY</td>
<td>Daily intervals used with datetime values</td>
<td>Each day</td>
</tr>
<tr>
<td>DTWEEK</td>
<td>Weekly intervals of seven days used with datetime values</td>
<td>Each day</td>
</tr>
<tr>
<td>DTWEEKDAYdaysW</td>
<td>Daily intervals with Friday-Saturday-Sunday counted as the same day (five-day work week with a Saturday-Sunday weekend.) days identifies the individual weekend days by number (1=Sunday ... 7=Saturday.) By default, days=&quot;17&quot;, so the default interval is DTWEEKDAY17W. This interval is only used with datetime values.</td>
<td>Each day</td>
</tr>
<tr>
<td>DTTENDAY</td>
<td>Ten-day intervals (a U.S. automobile industry convention) used with datetime values</td>
<td>1st, 11th, and 21st of each month</td>
</tr>
<tr>
<td>DTSEMIMONTH</td>
<td>Half-month intervals used with datetime values</td>
<td>1st and 16th of each month</td>
</tr>
<tr>
<td>DTMONTH</td>
<td>Monthly intervals used with datetime values</td>
<td>1st of each month</td>
</tr>
<tr>
<td>DTQTR</td>
<td>Quarterly (three-month) intervals used with datetime values</td>
<td>1-Jan, 1-Apr, 1-Jul, 1-Oct</td>
</tr>
<tr>
<td>DTSEMIYEAR</td>
<td>Semiannual (six-month) intervals used with datetime values</td>
<td>1-Jan, 1-Jul</td>
</tr>
<tr>
<td>DTYEAR</td>
<td>Yearly intervals used with datetime values</td>
<td>1-Jan</td>
</tr>
<tr>
<td>DTSECOND</td>
<td>Second intervals used with datetime values</td>
<td>Seconds</td>
</tr>
<tr>
<td>DTMINUTE</td>
<td>Minute intervals used with datetime values</td>
<td>Minutes</td>
</tr>
<tr>
<td>DTHOUR</td>
<td>Hour intervals used with datetime values</td>
<td>Hours</td>
</tr>
<tr>
<td>SECOND</td>
<td>Second intervals used with time values</td>
<td>Seconds</td>
</tr>
<tr>
<td>MINUTE</td>
<td>Minute intervals used with time values</td>
<td>Minutes</td>
</tr>
<tr>
<td>HOUR</td>
<td>Hourly intervals used with time values</td>
<td>Hours</td>
</tr>
</tbody>
</table>

These may seem quite self-explanatory, but if you just jump right in and use these intervals without reading about them first, you may not end up with what you expected.
WHEN IS A YEAR NOT A YEAR?

The syntax for the INTCK() function is: \texttt{INTCK(interval,start-of-period,end-of-period)}; \texttt{interval} must be in quotes. This function calculates the number of \textit{intervals} between the dates represented by \textit{start-of-period} and \textit{end-of-period}. To illustrate the potential for problems with SAS intervals, consider the following program:

```sas
DATA _NULL_;  
v1 = INTCK('YEAR','01jan2005'd,'01jan2006'd);  
v2 = INTCK('YEAR','31dec2005'd,'01jan2006'd);  
PUT v1= +3 v2= +3;  
RUN;  
v1=1    v2=1
```

Now wait a minute. We know that a year from December 31, 2005, is not January 1, 2006. What happened? SAS intervals are not a shortcut to doing the math. The INTCK() function counts the number of times that the period \textit{interval} begins between \textit{start-of-period} and \textit{end-of-period}. It does not count the number of complete intervals between \textit{start-of-period} and \textit{end-of-period}. Therefore, for any date in 2005, the first year period starts on January 1, 2005. Therefore, given any starting date in 2005, the number of YEAR intervals INTCK counts will be one for any given ending date in 2006. There's enormous potential for bad results if you misunderstand how INTCK() calculates. The important thing to remember about intervals is that they are based on the beginning point of the interval, and not the beginning date given to the function.

The INTNX() function advances a given date, time or datetime by a specified number of intervals. The syntax for this function is \texttt{INTNX(interval,start-date,number-of-increments,alignment)}; where \texttt{interval} is one of the SAS intervals from the previous page (again in quotes), \texttt{start-date} is the starting date, and \texttt{number-of-increments} is how many intervals you want to change the date. \texttt{alignment} will adjust the result of INTNX() relative to the interval given. It can be "B", "M" or "E" (quotes necessary) for beginning (of the interval), middle, or end, respectively. In SAS 9 and above, the "S" operator will adjust the result to the same day as given in the \texttt{start-date} argument. To illustrate how INTNX() and alignment works, here is a sample program that adds six months to March 20, 2008, with the result in bold.

```sas
DATA _NULL_;  
a = INTNX('MONTH','20MAR2008'd,6);  
b = INTNX('MONTH','20MAR2008'd,6,'B');  
c = INTNX('MONTH','20MAR2008'd,6,'M');  
d = INTNX('MONTH','20MAR/2008'd,6,'E');  
e = INTNX('MONTH','20MAR/2008'd,6,'S');  
PUT '6 months from 3/20/2008 with default alignment = ' a mmddyy10.;  
PUT '6 months from 3/20/2008 aligned with beginning of MONTH interval= ' b mmddyy10.;  
PUT '6 months from 3/20/2008 aligned with middle of MONTH interval= ' c mmddyy10.;  
PUT '6 months from 3/20/2008 aligned with end of MONTH interval= ' d mmddyy10.;  
PUT '6 months from 3/20/2008 aligned with same day in MONTH interval= ' e mmddyy10.;  
RUN;  
```

BUT I DON'T WANT MY YEAR TO START ON JANUARY 1

Since the default starting point of an interval is at the beginning of it, the SAS System seems to have a blind spot when it comes to figuring out intervals that do not coincide with that. There is a way to shift the starting point of any given interval by creating your own interval definition. For example, what if you wanted to know the number of YEAR intervals between two dates, but instead of calculating calendar years, you wanted to calculate your company's fiscal year, which starts on February 1? You tell SAS how many periods to shift. Each interval has a shift unit. For years, the shift unit is months, so you will tell SAS to shift the starting point of the year in terms of months. A shifted interval is the interval name, followed by a period, and the number of periods to shift. To shift the start of the YEAR interval to February 1, you would use the interval "YEAR.2".

It is important to remember, when you count the number of periods to shift, you need to include the beginning of the period. We are not shifting the start of the year interval by one month to move to February; we are moving the start of the year to the second month. If you do not account for this, you will be off by one unit. Another handy way to think of it is that the YEAR.1 interval is the same as the YEAR interval.
ISO 8601 AND SAS DATES, TIMES, AND DATETIME VALUES

For us long-time SAS users (pre-version 9), making dates, times and datetimes adhere to the ISO 8601 standard required a little bit of ingenuity. There are macros that have been constructed to take the ISO representations and turn them into SAS-appropriate values, and to transform SAS values for date, time, and datetimes into the ISO 8601 representation. If you are using one of those macros, and your company now has SAS version 9 or higher (the ISO 8601 formats and informats were first introduced in SAS 8.2, but version 9 has several enhancements), you can stop now. For the sake of brevity, this paper will only discuss the conversion of dates, times, and datetimes between SAS and their ISO 8601 representations, and not the slightly more complicated concept of describing periods.

To convert an ISO 8601 date value (which is a character variable) into its SAS equivalent and its components, you just need three lines of code:

\[
\begin{align*}
\text{SAS-datetime-variable} &= \text{INPUT(COMPRESS(ISO 8601-date-variable,\'-.\'),B8601DT.);} \\
\text{SAS-time-variable} &= \text{TIMEPART(SAS-datetime-variable);} \\
\text{SAS-date-variable} &= \text{DATEPART(SAS-datetime-variable);}
\end{align*}
\]

That's it, unless the ISO 8601 date value includes time zones, in which case the date time conversion becomes:

\[
\text{SAS-datetime-variable} = \text{INPUT(COMPRESS(ISO 8601-date-variable,\'-.:+\'),B8601DZ.);}
\]

Why do you need to use the COMPRESS() function? It's a solution to a problem described in SAS Note 43337. If you have a partial ISO date value (e.g., “2011-03-20T15:30”), trying to process it with the E8601DT. informat will not work, because the default length of the informat is 19 (which is also the minimum length allowed by SAS), and the length of the character value is 16. Therefore, you cannot specify E8601DT16. (if you try, you will get an error.) Conversely, if you use the default length, you will get the error message: “NOTE: Invalid argument to function INPUT” in your SAS log.

It is even easier to represent a SAS datetime value using the ISO 8601 standard: you simply use one of the following formats:

- **B8601DA.** Represents a SAS date value in the format \( \text{yyyyymmdd} \)
- **B8601DT.** Represents a SAS datetime value in the format \( \text{yyyyymmddThhmmssffffff} \) (\( \text{ffffff} \) represents fractional seconds)
- **B8601DZ.** Represents a SAS datetime value in the format \( \text{yyyyymmddThhmmss+-hhmm} \)
- **B8601TM.** Represents a SAS time value in the format \( \text{hhmmssffffff} \) (\( \text{ffffff} \) represents fractional seconds)

If you want delimiters in your ISO 8601 values (e.g., “:”, “-”)—which are the characters we eliminated from the ISO date values using the COMPRESS() function in the conversion examples above—then use the corresponding E8601 formats.

EXCEL

We cannot overlook the potential for incorrect results whenever you convert from Microsoft Excel into SAS or vice versa. The most likely issue comes from old dates again. Microsoft Excel uses a zero point of January 1, 1900, but unlike the SAS System, it cannot count backwards from zero. Therefore, any date prior to 1900 is represented as text, not as an Excel date value. What do you do if you have dates before January 1, 1900, to convert? Going to Excel, you will have to store them as character strings, and you will not be able to use any of the Excel math functions on them. On the other hand, if you are going from Excel to SAS, then you can use an INPUT statement and a DATA step to process a CSV file. If you use the IMPORT wizard and/or procedure, It will give you the dreaded dot, or possibly record the variable as a character string, neither of which will work for an actual date value in your SAS dataset.

SUMMARY

The way SAS handles dates, times, and datetimes allows for a great deal of flexibility in the display, manipulation, and computation of these important pieces of data. However, that flexibility can also make it easy to make mistakes. A solid basic understanding of dates and times in SAS is necessary to use this powerful tool correctly. Dates, times, and datetimes are stored as numbers in the SAS system. Dates are counted in days from a zero point of January 1, 1960. Times are counted in seconds from a zero point of midnight of the current day, and datetimes are counted in seconds since midnight, January 1, 1960. Each day that passes increments the day counter by one, and each second that passes increments the time and datetime counters by one.

SAS dates and times are displayed by using formats. There are many built-in formats in SAS. Each of these displays a date, time, or a datetime value as something we can recognize. Date formats are specific to date values, time formats are specific to time values, and datetime formats are specific to datetime values. If you use the wrong type of format, your display will be incorrect. If none of the built-in formats meets your needs, you can create your own, either by defining the display to correlate with certain values, or by defining the display with a pictorial template using...
symbolic directives. You can use the formats to customize titles and footnotes in your output with date and time information.

To translate dates and times as we know them into the values recognized by SAS, we use informats. As with formats, there are many built-in informats designed to translate standard alphanumeric or numeric representations of dates into SAS date, time, or datetime values. You need to specify the number of characters to be processed by the informat, and you should use the correct informat for the alphanumeric characters you are processing. If you do not know the format of the alphanumeric characters you will be processing, you can use the ANYDATE informats, but be prepared for your job to take longer.

You may extract the individual components of a SAS date, time, or datetime value, such as month number, day or hour. However, you need to make sure you do not try to extract a month from a time, or an hour from a date. You may also create a SAS date, time, or datetime value from individual components. Whenever you do this, remember that all of the arguments to the MDY(), HMS(), and DHMS() functions must be non-missing for these functions to work.

It is frequently useful to refer to periods of time by familiar references such as year or quarter. The SAS System handles this with intervals. There are several such standard periods defined in SAS, and two functions that use them, INTCK() and INTNX(). The most important thing to remember is that these functions use SAS intervals by measuring the intervals from the start point of the interval, not the dates supplied to the functions as arguments.

There are many more uses of SAS dates, more functions relating to dates and times and datetimes, and more capabilities of intervals than have been shown in this brief paper. This introduction was written to provide an overview of the basic ways to handle and manipulate dates and times with the SAS system to address a large percentage of date- and time-related tasks. Should you need more than what is in this paper, the combination of the flexibility of the SAS System with respect to dates and times and the tools it provides you to handle them should allow you to accomplish your tasks.

REFERENCES


CONTACT INFORMATION:

Further inquiries are welcome to:
Derek Morgan
Covidien, Pharmaceuticals
675 McDonnell Blvd.
Biostatistics, 30-1
Hazelwood, MO 63042
E-mail: derek.morgan@covidien.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. © indicates USA registration. Other brand and product names are trademarks of their respective companies.